$\mathbb{R}_5[X]$ est de dimension finie, donc toutes les normes y sont équivalentes, en particulier N_∞ et la norme N associée au produit scalaire.

Par contre, $\forall n \in \mathbb{N}^*,\ N(X^n) = \sqrt{\dfrac{2}{2n+1}}$ et $N_\infty(X^n) = 1$; $\dfrac{N}{N_\infty}$ n'est pas bornée sur $\mathbb{R}[X] \setminus \{0\}$ donc N_∞ et N ne sont pas équivalentes sur $\mathbb{R}[X]$.

O20-102

```
> restart;
> ps:=(P,Q)->int(P*Q,t=-1..1);nor:=P->sqrt(ps(P,P));
```

$$ps := (P, Q) \to \int_{-1}^{1} P\, Q\, dt$$

$$nor := P \to \sqrt{\mathrm{ps}(P, P)}$$

```
> base:=[seq(t^(i-1),i=1..6)];
```

$$base := [1, t, t^2, t^3, t^4, t^5]$$

```
> m:=(i,j)->ps(base[i],base[j]);
```

$$m := (i, j) \to \mathrm{ps}(base_i, base_j)$$

```
> with(LinearAlgebra):
> M:=Matrix(6,6,m);
```

$$M := \begin{bmatrix} 2 & 0 & \frac{2}{3} & 0 & \frac{2}{5} & 0 \\ 0 & \frac{2}{3} & 0 & \frac{2}{5} & 0 & \frac{2}{7} \\ \frac{2}{3} & 0 & \frac{2}{5} & 0 & \frac{2}{7} & 0 \\ 0 & \frac{2}{5} & 0 & \frac{2}{7} & 0 & \frac{2}{9} \\ \frac{2}{5} & 0 & \frac{2}{7} & 0 & \frac{2}{9} & 0 \\ 0 & \frac{2}{7} & 0 & \frac{2}{9} & 0 & \frac{2}{11} \end{bmatrix}$$

```
> baseo[1]:=base[1]/nor(base[1]);
```

$$baseo_1 := \frac{\sqrt{2}}{2}$$

```
> for i from 2 to 6 do
  v:=base[i]-sum('ps(base[i],baseo[k])*baseo[k]','k'=1..i-1);
  baseo[i]:=v/nor(v)
  od;
```

$$v := t$$

$$baseo_2 := \frac{t\sqrt{6}}{2}$$

$$v := t^2 - \frac{1}{3}$$

$$baseo_3 := \frac{3\left(t^2 - \frac{1}{3}\right)\sqrt{10}}{4}$$

$$v := t^3 - \frac{3}{5}t$$

$$baseo_4 := \frac{5\left(t^3 - \frac{3}{5}t\right)\sqrt{14}}{4}$$

$$v := t^4 + \frac{3}{35} - \frac{6}{7}t^2$$

$$baseo_5 := \frac{105\left(t^4 + \frac{3}{35} - \frac{6}{7}t^2\right)\sqrt{2}}{16}$$

$$v := t^5 + \frac{5}{21}t - \frac{10}{9}t^3$$

$$baseo_6 := \frac{63\left(t^5 + \frac{5}{21}t - \frac{10}{9}t^3\right)\sqrt{22}}{16}$$

```
> plot([seq(baseo[k],k=1..6)],t=-1..1);
```


2
1
-1
-2
-1
-0.8
-0.6
-0.4
-0.2
0.2
0.4
0.6
0.8
1
t

>